


Gas Turbine Efficiency

▼ Introduction

This application plots the efficiency of a gas turbine (based upon an ideal Brayton cycle) as a function of the compressor pressure ratio.

A procedure is developed that computes the efficiency as a function of the compressor inlet temperature and pressure, the combustor outlet temperature, and the compressor pressure ratio; any of these parameters can be modified and explored.

The working fluid is air, whose thermodynamic properties of air are provided by the [ThermophysicalData](#) package



▼ Efficiency

```
> restart:
> turbineEfficiency := proc(Tmin, Tmax, Pmin, r)
```

```

    # Tmin and Pmin are the compressor inlet temperature and
pressure
    # Tmax is the combustor outlet temperature
    # r is the compressor pressure ratio

    uses ThermophysicalData:
    local P1, T1, h1, s1, s2, P2, T2, h2, P3, T3, h3, s3, P4, s4,
T4, h4, eta:

    # Compressor inlet (isentropic)
    P1 := Pmin:
    T1 := Tmin:
    h1 := Property(enthalpy, temperature = T1, pressure = P1,
air):
    s1 := Property(entropy, temperature = T1, pressure = P1, air)
:

    # Combustor inlet (isobaric)
    s2 := s1:
    P2 := r*P1:
    T2 := Property(temperature, pressure = P2, entropy = s2, air)
:
    h2 := Property(enthalpy, temperature = T2, pressure = P2,
air):

    # Turbine inlet
    P3 := P2:
    T3 := Tmax:
    h3 := Property(enthalpy, temperature = T3, pressure = P3,
air):
    s3 := Property(Smass, T = T3, P = P3, air):

    # Turbine exit (isentropic)
    P4 := P1:
    s4 := s3:
    T4 := Property(temperature, pressure = P4, entropy = s4, air)
:
    h4 := Property(enthalpy, temperature = T4, pressure = P4,
air):


    # Thermal efficiency
    eta := 1 - (h4 - h1) / (h3 - h2):

    return eta

end proc:
> plot(turbineEfficiency(773, 1573, 101325, r), r = 1..8, labels =
["Pressure Ratio", "Efficiency"], labeldirections = [horizontal,
vertical], font = [Calibri], labelfont = [Calibri], title =
"Efficiency of a Gas Turbine", titlefont = [Calibri, 20],
gridlines, size = [800,500])

```

Efficiency of a Gas Turbine



Legal Notice: © Maplesoft, a division of Waterloo Maple Inc. 2017. Maplesoft and Maple are trademarks of Waterloo Maple Inc. This application may contain errors and Maplesoft is not liable for any damages resulting from the use of this material. This application is intended for non-commercial, non-profit use only. Contact Maplesoft for permission if you wish to use this application in for-profit activities.